Qspice - Model Generators Guide by KSKelvin

KSKelvin Kelvin Leung

Created on : 10-29-2024
Last Update : 6-29-2025

Model Generator and Precaution in using this Guide

- Model Generator
  - Model generators are in File > Model Generator > Diode/JFET/MOSFET
  - Execute one of these model generators, within the subprogram, it has official HELP



- Precaution in using this Guide
  - The model generator appears to still be subject to change. If you are unable to replicate the example provided in this guideline, it may be related to a change in the model generator
  - I cannot guarantee the accuracy of this guideline as it heavily relies on parameter studies through these model generators. This guideline is still in its preliminary status

Technique in Digitizing Datasheet

Digitize with Crosshair Cursor and Arrow Slight Adjustment

Step #2 : **[Crosshair Cursor]**

- Move cursor to this area, **hold** Left mouse button
  Now, the cursor become a crosshair



Figure 2. Forward Current vs. Forward Voltage



Step #1

- Click to begin digitize procedure

Step #3 :

- Move crosshair cursor to pdf to digitize lower left corner
  (Can **use arrow key to adjust crosshair position precisely**)
- Release left mouse button and location is digitized
- [Repeat Step #2 and #3 until all points is digitized]

Useful Technique in using Digitize

To digitize a very small value on a linear scale, use the arrow keys to make fine adjustments to the cursor. Monitor the y-axis value to ensure it is one digit less than the y-axis value of the LowerLeft.



Diode Model Generator

DIODE.exe

Diode Model Generator

Parameters Generation

Diode Model Generator – Preliminaries Tab

Determine : mfg, Iave, Vrev, Eg, tt, Cjo**, BV, IBV, NBV



mfg (display only)

Iave (display only)

Vrev (display only)

$$\mathbf{tt} = \frac{\text{Rev Recovery Charge}}{\text{I @ Rev Recovery}}$$

Cjo **

BV

IBV

Calculate **NBV** (formula unknown)

Silicon	**Eg** = 1.11
Schottky Barrier Diode	**Eg** = 0.69
Germanium	**Eg** = 0.67
Gallium Nitride(GaN)	**Eg** = 3.47
Siconon Carbide(SiC)	**Eg** = 3.26
Gallium Arsenide(GaAs)	**Eg** = 1.42

BV, **IBV**, **NBV** will be generated if this box is non-zero

Cjo : This only determine Cjo in I-V Curve digitized tab. If Capacitance digitized tab is used, this Cjo will be ignored

Diode Model Generator – I-V Curve Tab

Determine : Is, N, Rs, TRs1, XTI



Determine **Is**, **N, Rs**

Determine **TRs1, XTI**
If Temp set to be identical nominal temperature, force TRs1=0 and XTI=3
(If XTI=3, TRs1 may not generated)

** where to digitize point #2 can affect Is and N
If point #1 and #2 are closer (e.g. at #2a), Is tends to be calculated smaller

Diode Model Generator – Capacitance Tab

Determine : Cjo, m, Vj



Determine **Cjo**, **m** and **Vj**

Diode Model Generator

Example – Datasheet of Onsemi MURS1200

Example – Onsemi MURS120 Datasheet to Model Generator

MAXIMUM RATINGS

Rating	Symbol	MURS/SURS8/NRVUS 105T3	110T3	115T3	120T3
Peak Repetitive Reverse Voltage Working Peak Reverse Voltage DC Blocking Voltage	V_{RRM} V_{RWM} V_R	50	100	150	200
Continuous Forward Current	$I_{F(DC)}$	1.0 @ T_L = 159°C 2.0 @ T_L = 139°C			



Maximum Reverse Recovery Time (i_F = 1.0 A, di/dt = 50 A/µs, V_R = 30 V) (i_F = 0.5 A, i_R = 1.0 A, I_R to 0.25 A)	t_{rr}	35 25

Reverse Recovery Time is given at IF=1A

- I @ Rev. Recovery = 1A
- Rev. Recovery Charge = trr * IF = 35n



Figure 3. Typical Capacitance

Example – Onsemi MURS120 Datasheet to Model Generator



Figure 1. Typical Forward Voltage





Example – Onsemi MURS120 Datasheet to Model Generator



Figure 3. Typical Capacitance





Diode Model Generator

Example – Datasheet of Microchip APT15D100BCT

Example – Microchip APT15D100BCT Datasheet to Model Generator

MAXIMUM RATINGS

Rating	Symbol	MURS/SURS8/NRVUS			
		105T3	110T3	115T3	120T3
Peak Repetitive Reverse Voltage Working Peak Reverse Voltage DC Blocking Voltage	V_{RRM} V_{RWM} V_R	50	100	150	200
Continuous Forward Current	$I_{F(DC)}$	1.0 @ T_L = 159°C 2.0 @ T_L = 139°C			



t_{rr}	Reverse Recovery Time	I_F = 15A, di_F/dt = -200A/µs V_R = 667V, T_C = 25°C	-	260		ns
Q_{rr}	Reverse Recovery Charge		-	540		nC
I_{RRM}	Maximum Reverse Recovery Current		-	4	-	Amps



Figure 8. Junction Capacitance vs. Reverse Voltage

Example – Microchip APT15D100BCT Datasheet to Model Generator

Qspice reference is log plot (y-axis log, x-axis linear), marker 1-5 is different if linear plot is used



Figure 2. Forward Current vs. Forward Voltage

Marker 1 to 5 if linear plot is used (y-axis and x-axis are both linear)





Example – Microchip APT15D100BCT Datasheet to Model Generator



Figure 8. Junction Capacitance vs. Reverse Voltage





MOSFET Model Generator

MOSFET Model Generator

Parameters Generation

MOSFET Model Generator – Preliminaries Tab

Determine : mfg, Ids, Vds, Qg, Rg, Rds, Rd, Rs, Eg, tt

Rd+Rs = Rds(on) × (Rext ÷ (Rext + Rchannel))

- Rext = Rd + Rs (e.g. bond wire, contact)
- Rchannel is Rds of MOSFET channel

Not provide by datasheet, from Mike Engelhardt experience, in ~75% of resistance is contributed by Rext



mfg (display only)

Ids (display only)

Vds (display only)

Qg (display only)

Rg

Rds (display only), but also use to calculate **Rd** and **Rs**

Contribute to Calculation of **RonX** (formula unknown)

Rd = Rds(on) × (Rext ÷ (Rext + Rchannel)) × (Rd ÷ (Rd + Rs))

Rs = Rds(on) × (Rext ÷ (Rext + Rchannel)) − Rd

Not provided by datasheet, from Mike experience, Ratio of Rd and Rs from Rext, approximate by 50%

Eg

$$\textbf{tt} = \frac{\text{Rev Recovery Charge}}{\text{I @ Rev Recovery}}$$

Contribute to Calculation of **Cjo**

Representation:

$$\frac{x}{x+y} \rightarrow 1 : x \gg y$$

$$\frac{x}{x+y} \rightarrow 0 : y \gg x$$

MOSFET Model Generator – Output Characteristics

Determine : Vto, Kp, Lambda, RonX, eta, Vtotc



Determine **Vto**, **Kp, Lambda, RonX**

** **Rds,on** in preliminaries tab can be used to fine tuning Id vs Vds curve at lower Vds region

** **ETA** and **Vtotc** seems to be fixed
ETA=75m
Vtotc=-2m

MOSFET Model Generator – Body Diode

Determine : Is, N, Rs, TRs1, XTI



Determine **Is**, **N, Rs**

Determine **TRs1, XTI**
If Temp set to be identical nominal temperature, force TRs1=0 and XTI=3

** tt and Eg is from Perliminary tab
** Rs is series resistor in diode model, this will rename to Rb in VDMOS model in Gate Charge tab **(Rs → rename to Rb)**

MOSFET Model Generator – Gate Charge

Determine : Cgs, Cgdmin, Cgdmax, Cjo



mainly determine : **Cgs**

mainly determine : **Cgdmin**, **Cgdmax**

Affect **Cgs**, **Cgdmin**, **Cgdmax**, **Cjo**

No effect

Cjo=Zero-biased Output Cap (Preliminaries) – Cgdmax (min. value as 0)
[Cjo is body diode zero-bias capacitance]

MOSFET Model Generator

Example – Datasheet of Vishay IRF630

Example – Vishay IRF630 Datasheet to Model Generator

Gate input resistance	R_g	f = 1 MHz, open drain	0.6	-	3.3	Ω



Total gate charge	Q_g	V_{GS} = 10 V	I_D = 5.9 A, V_{DS} = 160 V, see fig. 6 and 13 [b]	-	-	43	nC
Gate-source charge	Q_{gs}			-	-	7.0	
Gate-drain charge	Q_{gd}			-	-	23	

Drain-source on-state resistance	$R_{DS(on)}$	V_{GS} = 10 V	I_D = 5.4 A [b]	-	-	0.40	Ω

Body diode reverse recovery time	t_{rr}	T_J = 25 °C, I_F = 5.9 A, dI/dt = 100 A/μs	-	170	340	ns
Body diode reverse recovery charge	Q_{rr}		-	1.1	2.2	nC

Coss at 0Vdc



Example – Vishay IRF630 Datasheet to Model Generator



Fig. 1 - Typical Output Characteristics, T_C = 25 °C



Example – Vishay IRF630 Datasheet to Model Generator



Fig. 7 - Typical Source-Drain Diode Forward Voltage





Example – Vishay IRF630 Datasheet to Model Generator



Fig. 6 - Typical Gate Charge vs. Gate-to-Source Voltage





According to Mike, Point #1 and #2 are sampled slightly lower and slightly higher than flat region

MOSFET Model Generator

Effect of Model Parameters

Output Characteristic (Rs, Rd)

Qspice : Sensitivity Study - Output Characteristic.qsch




```
.dc V1 0.1 50 1m V2 list 4.5 5
.model IRF630 VDMOS Rs=0.15*chg Rd=0.15
+ Vto=4.08 Kp=5.63 Lambda=3.85m
+ RonX=0.348 eta=75m Vtotc=-2m tt=0.186

;multiple chg (*chg)
.step param chg list 0.5 1 1.5
```




```
.dc V1 0.1 50 1m V2 list 4.5 5
.model IRF630 VDMOS Rs=0.15 Rd=0.15*chg
+ Vto=4.08 Kp=5.63 Lambda=3.85m
+ RonX=0.348 eta=75m Vtotc=-2m tt=0.186

;multiple chg (*chg)
.step param chg list 0.5 1 1.5
```





Output Characteristic (Vto, Kp, Lambda)

Qspice : Sensitivity Study - Output Characteristic.qsch




```
.dc V1 0.1 50 1m V2 list 4.5 5
.model IRF630 VDMOS Rs=0.15 Rd=0.15
+ Vto=4.08*chg Kp=5.63 Lambda=3.85m
+ RonX=0.348 eta=75m Vtotc=-2m tt=0.186

;multiple chg (*chg)
.step param chg list 0.98 1 1.02
```




```
.dc V1 0.1 50 1m V2 list 4.5 5
.model IRF630 VDMOS Rs=0.15 Rd=0.15
+ Vto=4.08 Kp=5.63*chg Lambda=3.85m
+ RonX=0.348 eta=75m Vtotc=-2m tt=0.186

;multiple chg (*chg)
.step param chg list 0.8 1 1.2
```




```
.dc V1 0.1 50 1m V2 list 4.5 5
.model IRF630 VDMOS Rs=0.15 Rd=0.15
+ Vto=4.08 Kp=5.63 Lambda=3.85m*chg
+ RonX=0.348 eta=75m Vtotc=-2m tt=0.186

;multiple chg (*chg)
.step param chg list 0.5 1 1.5
```







Bode Diode (Is, N, Rs)

Qspice : Sensitivity Study - Body Diode.qsch




```
.dc V1 0.5 1.5 1m
;.temp 25 150
.model IRF630 D Is=785p*chg N=1.33849 Rs=21m Eg=1.11
+ XTI=4.53261 Tnom=25 trs1=3.56m tt=0.186
+ mfg="Vishay" Vrev=200 Iave=9

;multiple chg (*chg)
.step param temp list 25 150 param chg list 0.5 1 1.5
```




```
.dc V1 0.5 1.5 1m
;.temp 25 150
.model IRF630 D Is=785p N=1.33849*chg Rs=21m Eg=1.11
+ XTI=4.53261 Tnom=25 trs1=3.56m tt=0.186
+ mfg="Vishay" Vrev=200 Iave=9

;multiple chg (*chg)
.step param temp list 25 150 param chg list 0.8 1 1.2
```




```
.dc V1 0.5 1.5 1m
;.temp 25 150
.model IRF630 D Is=785p N=1.33849 Rs=21m*chg Eg=1.11
+ XTI=4.53261 Tnom=25 trs1=3.56m tt=0.186
+ mfg="Vishay" Vrev=200 Iave=9

;multiple chg (*chg)
.step param temp list 25 150 param chg list 0.8 1 1.2
```







Bode Diode (XTI, trs1) : Temperature Effect

Qspice : Sensitivity Study - Body Diode.qsch




```
.dc V1 0.5 1.5 1m
;.temp 25 150
.model IRF630 D Is=785p N=1.33849 Rs=21m Eg=1.11
+ XTI=4.53261*chg Tnom=25 trs1=3.56m tt=0.186
+ mfg="Vishay" Vrev=200 Iave=9

;multiple chg (*chg)
.step param temp list 25 150 param chg list 0.5 1 1.5
```




```
.dc V1 0.5 1.5 1m
;.temp 25 150
.model IRF630 D Is=785p N=1.33849 Rs=21m Eg=1.11
+ XTI=4.53261 Tnom=25 trs1=3.56m*chg tt=0.186
+ mfg="Vishay" Vrev=200 Iave=9

;multiple chg (*chg)
.step param temp list 25 150 param chg list 0.5 1 1.5
```





Gate Charge (Cgs, Cgdmin, Cgdmax)

Qspice : Sensitivity Study - Gate Charge.qsch

Ggs




```
;multiple chg (*chg)
.step param chg list 0.5 1 1.5
.tran 0.05
.ic V(G)=0
.model IRF630 VDMOS Rs=0.15 Rd=0.15 Rg=2 Vto=4.08 Kp=5.63
+ lambda=3.85m RonX=0.348 eta=75m Vtotc=-2m Is=785p N=1.33849
+ Rb=21m Eg=1.11 XTI=4.53261 trb1=3.56m Cgs=668p*chg Cgdmin=133p
+ Cgdmax=1.57n Cjo=0 tt=0.186 Tnom=25 mfg="Vishay" Vds=200
+ Ids=9 Ron=0.4 Qg=43n
```



Ggdmin




```
;multiple chg (*chg)
.step param chg list 0.5 1 1.5
.tran 0.05
.ic V(G)=0
.model IRF630 VDMOS Rs=0.15 Rd=0.15 Rg=2 Vto=4.08 Kp=5.63
+ lambda=3.85m RonX=0.348 eta=75m Vtotc=-2m Is=785p N=1.33849
+ Rb=21m Eg=1.11 XTI=4.53261 trb1=3.56m Cgs=668p Cgdmin=133p*chg
+ Cgdmax=1.57n Cjo=0 tt=0.186 Tnom=25 mfg="Vishay" Vds=200
+ Ids=9 Ron=0.4 Qg=43n
```



Ggdmax




```
;multiple chg (*chg)
.step param chg list 0.5 1 1.5
.tran 0.05
.ic V(G)=0
.model IRF630 VDMOS Rs=0.15 Rd=0.15 Rg=2 Vto=4.08 Kp=5.63
+ lambda=3.85m RonX=0.348 eta=75m Vtotc=-2m Is=785p N=1.33849
+ Rb=21m Eg=1.11 XTI=4.53261 trb1=3.56m Cgs=668p Cgdmin=133p
+ Cgdmax=1.57n*chg Cjo=0 tt=0.186 Tnom=25 mfg="Vishay" Vds=200
+ Ids=9 Ron=0.4 Qg=43n
```



Waveform Viewer Color Configuration for Sensitivity Study

- In sensitivity study, as triple sweep is used in Output Characteristic and Body Diode, color trace in waveform viewer is setup as
  - Data Trace 1 Color : [0,255,0]
  - Data Trace 2 Color : [0,255,0]
  - Data Trace 3 Color : [0,63,255]
  - Data Trace 4 Color : [0,63,255]
  - Data Trace 5 Color : [255,63,0]
  - Data Trace 6 Color : [255,63,0]



MOSFET Model Generator

Example – Recreate from a model

Determine Rds(on), Vgs, Idrain @ Rds(on) and Rext÷(Rext+Rchannel)

Qspice : Preliminaries (Rdson Vgs Idrain and Rext).qsch

- Rext : Rd + Rs
  - Rds(on) is basically consist of Rext (external resistance : Rd, Rs) and Rchannel (channel resistance)
  - To estimate Rext, fully turn ON a FET model with extreme gate-source voltage, which minimized Rchannel and Rds(on) is dominated by Rext
  - In this example, Rs+Rd=Rext=39.8mΩ. And by extreme gate-source, Rext=40.6mΩ




```
; force Rext >> Rchannel with extreme Vgs
;   therefore, Rext = Rd + Rs
.param Idrain=1

.dc Vgs 5 50 .1 ; sweep to unrealistically high for Rext
.func Rds() V(d)/I(Id)
.plot Rds()

.meas Rdson max Rds()
.meas Ids@Rdson min I(Id)
.meas Vgs@Rdson min V(g)
.meas Rext min Rds()
.meas Reqn Rext/(Rext+(Rdson-Rext))
```





P/N	Manufacturer	Vds[V]	Ids[A]	Rds(on)[mΩ]	Qg[nC]
BSC123N08NS3	Infineon	80	55	12.3	19.0
EPC2218	EPC	100	60	2.4	10.5
GS66508B	GaN Systems	650	30	50.0	6.1
GS66516T	GaN Systems	650	60	25.0	14.2
AO4262E	Alpha & Omega	60	16	6.6	15.0
DMZ6005	ARK Microelectronics Co.	600	0	700000.0	1.6
FTA07N60	ARK Microelectronics Co.,	600	7	900.0	38.6
SQ3426EEV	Vishay	60	7	57.0	7.6
UF3C065030	Qorvo	650	65	27.0	51.0

```
.model SQ3426EEV VDMOS Rs=19.9m Rd=19.9m Rg=2.4 Vto=2.76 Kp=14.7
+ lambda=0.123 RonX=2.3 eta=.1 Is=65.1p N=1.22362 Rb=11.4m
+ Eg=1.11 XTI=2.60857 trb1=1.64m Cgs=700p Cgdmin=40p Cjo=200p
+ Cgdmax=700p mfg=Vishay Vds=60 Ids=7 Ron=57m Qg=7.6n
```

- Now put,
  - Rds(on) = 53.2mΩ
  - Vgs @ Rds(on) = 5V
  - Idrain @ Rds(on) = 1A
  - Rext = Rs+Rd = 40.6mΩ
  - Rchannel@Rds(on)
    = Rds(on) – Rext
    = 53.2mΩ – 40.6mΩ
    = 12.6mΩ
  - Rext÷(Rext+Rchannel)
    = 40.6Ω /(40.6Ω +12.6Ω)
    =0.763Ω

Determine Rg from a MOSFET Model

Qspice : Preliminaries - Rg.qsch

- Rg
  - Rg is series resistance in gate
  - Rg can be identified with ac analysis and only read the real part with Cartesian representation
  - Now, put
    - Rg = value of Zr()




```
.ac list 1Meg
.func Zr() re(1/-I(Vgs))
.plot Zr()
.meas Rg param Zr()
```




MOSFET Selection Guide

Gate Voltage:

P/N	Manufacturer	Vds[V]	Ids[A]	Rds(on)[mΩ]	Qg[nC]
BSC123N08NS3	Infineon	80	55	12.3	19.0
EPC2218	EPC	100	60	2.4	10.5
GS66508B	GaN Systems	650	30	50.0	6.1
GS66516T	GaN Systems	650	60	25.0	14.2
AO4262E	Alpha & Omega	60	16	6.6	15.0
DMZ6005	ARK Microelectroncs Co.	600	0	700000.0	1.6
FTA07N60	ARK Microelectronics Co.,	600	7	900.0	38.6
SQ3426EEV	Vishay	60	7	57.0	7.6
UF3C065030	Qorvo	650	65	27.0	51.0

```
.model SQ3426EEV VDMOS Rs=19.9m Rd=19.9m Rg=2.4 Vto=2.76 Kp=14.7
+ lambda=0.123 RonX=2.3 eta=.1 Is=65.1p N=1.22362 Rb=11.4m
+ Eg=1.11 XTI=2.60857 trb1=1.64m Cgs=700p Cgdmin=40p Cjo=200p
+ Cgdmax=700p mfg=Vishay Vds=60 Ids=7 Ron=57m Qg=7.6n
```

More Info | Cancel | OK